Solving the heat equation

```
> EQ := diff( u(t,x), t ) = alpha*diff(u(t,x),x,x) ;
```

$$EQ := \frac{\partial}{\partial t} u(t,x) = \alpha \left(\frac{\partial^2}{\partial x^2} u(t,x) \right) \quad \textbf{(1)}$$

use separation if variable

```
> SEP := u(t,x)= T(t)*X(x) ;
```

$$SEP := u(t,x) = T(t)\, X(x) \quad \textbf{(2)}$$

Substituting

```
> subs( SEP, EQ ) ;
```

$$\frac{\partial}{\partial t} (T(t)\, X(x)) = \alpha \left(\frac{\partial^2}{\partial x^2} (T(t)\, X(x)) \right) \quad \textbf{(3)}$$

expanding

```
> EQ1 := expand(subs( SEP, EQ )) ;
```

$$EQ1 := \left(\frac{d}{dt} T(t) \right) X(x) = \alpha\, T(t) \left(\frac{d^2}{dx^2} X(x) \right) \quad \textbf{(4)}$$

lhs depends only on x, rhs depends only on t

```
> EQ2 := rhs(EQ1)/(X(x)*T(t)) = lhs(EQ1)/(X(x)*(T(t)));
```

$$EQ2 := \frac{\alpha \left(\frac{d^2}{dx^2} X(x) \right)}{X(x)} = \frac{\frac{d}{dt} T(t)}{T(t)} \quad \textbf{(5)}$$

2 new differentisal equations are deduced

```
> EQa := rhs(EQ2)=-lambda ;
  EQb := lhs(EQ2)=-lambda ;
```

$$EQa := \frac{\frac{d}{dt} T(t)}{T(t)} = -\lambda$$

$$EQb := \frac{\alpha \left(\frac{d^2}{dx^2} X(x) \right)}{X(x)} = -\lambda \quad \textbf{(6)}$$

```
> EQa*T(t) ; dsolve(%) ;
```

$$\frac{d}{dt} T(t) = -T(t)\, \lambda$$

$$T(t) = \_C1\, e^{-\lambda t} \quad \textbf{(7)}$$

```
> EQb*X(x) ; dsolve(%) ;
```

$$\alpha \left(\frac{d^2}{dx^2} X(x) \right) = -X(x)\, \lambda$$

$$X(x) = \_C1 \sin\left(\frac{\sqrt{\lambda}\, x}{\sqrt{\alpha}} \right) + \_C2 \cos\left(\frac{\sqrt{\lambda}\, x}{\sqrt{\alpha}} \right) \quad \textbf{(8)}$$

The general solution for the function of the form T(t)*X(x)

```
> U := exp(-lambda*t)*(c1*sin(sqrt(lambda)*x/sqrt(alpha))+c2*cos(sqrt
(lambda)*x/sqrt(alpha)));
```

$$U := e^{-\lambda t} \left(c1 \sin\left(\frac{\sqrt{\lambda}\, x}{\sqrt{\alpha}} \right) + c2 \cos\left(\frac{\sqrt{\lambda}\, x}{\sqrt{\alpha}} \right) \right) \quad \textbf{(9)}$$

```
> eval(subs(t=0,U)) ;
```

$$c1 \sin\left(\frac{\sqrt{\lambda}\, x}{\sqrt{\alpha}} \right) + c2 \cos\left(\frac{\sqrt{\lambda}\, x}{\sqrt{\alpha}} \right) \quad \textbf{(10)}$$

```
> Uk := simplify(subs(lambda=alpha*k^2,U)) assuming alpha>0,
k::integer, k>0;
```

$$Uk := e^{-\alpha k^2 t} \left(c1 \sin(k x) + c2 \cos(k x) \right) \quad \textbf{(11)}$$

```
> sum( subs(c1=c1(k),c2=c2(k),Uk),k=1..N) ; eval(subs(t=0,%)) ;
```

$$\sum_{k=1}^{N} e^{-\alpha k^2 t} \left(c1(k) \sin(k x) + c2(k) \cos(k x) \right)$$

$$\sum_{k=1}^{N} \left(c1(k) \sin(k x) + c2(k) \cos(k x) \right) \quad \textbf{(12)}$$